

# Proofs

Arthur Ryman, [arthur.ryman@gmail.com](mailto:arthur.ryman@gmail.com)

October 28, 2018

## Abstract

This article is a Z Notation specification for proofs and proof checking. It has been type checked by *f*UZZ. The definitions that appear here are taken from Lemmon's book, *Beginning Logic*. The purpose of this specification is to guide the development of a proof checker aimed at Z specifications

## 1 Introduction

For a long time I have thought that it would be extremely useful to be able to write formal proofs concerning the mathematical objects defined in Z specifications. There are some very mature proof assistants available. I know something about Coq, but unfortunately it's style of proof is very different that that one finds in mathematical papers. A Coq proof consists of a list of tactics that represent higher level aggregates of deductions. This makes sense for a proof assistant since its job is to help the user discover proofs.

While a proof assistant might make sense in some contexts, proper development of a mathematical paper consists of a gradual introduction of concepts and lemmas leading to the main results. The proofs should, in some sense, write themselves. The focus of a mathematical paper should be on explanation and clarity. The proofs should be easy to read. I'd therefore really like something that would let me write and check natural looking proofs.

In contrast to Coq, the style of proof presented by Lemmon is very clear and explicit. However, the task of checking such a proof could easily be delegated to a program, much the same way that *f*UZZ type checks Z.

I believe that the kernel of a proof checker could be very small. It is basically an engine driven by a set of deduction rules. The engine simply needs to check that each deduction rule gets applied correctly. Even if it turns out that writing such an engine is too much work, the exercise of developing at least a simple version should give me a greater appreciation of tools like Coq and enable me to use them more productively.

My plan of attack is to formalize the concept of proof as described by Lemmon, starting with the propositional calculus.

## 2 Propositions

The *propositional calculus* defines a set of formal *statements* or *propositions* without being concerned about the subject matter described by those statements. The only restriction on these statements is that, in any given context, they possess a *truth value* of either *true* or *false*.

### 2.1 Prop

A proposition is also referred to as a *well-formed formula* or *wff* for short. This terminology stems from the traditional development of the propositional calculus in terms of a language of sentences over an alphabet with rules that prescribe when a given sentence is *well-formed*. However, here we will dispense with the language viewpoint, and its associated parsing issues, and move directly to the end result of parsing, namely the creation of *abstract syntax trees* or *ASTs* for short.

Z notation has a convenient mechanism for specifying the structure of ASTs, namely that of *free types*. Let *Prop* denote the free type of all propositions of the propositional calculus. The free type definition of *Prop* is as follows.

$$\begin{aligned}
 \text{PropVar} &::= P \mid Q \mid R \mid S \mid T \\
 &\quad \mid (-') \langle\langle \text{PropVar} \rangle\rangle \\
 \text{Prop} &::= \text{true} \mid \text{false} \\
 &\quad \mid \text{var} \langle\langle \text{PropVar} \rangle\rangle \\
 &\quad \mid \neg \langle\langle \text{Prop} \rangle\rangle \\
 &\quad \mid (- \wedge -) \langle\langle \text{Prop} \times \text{Prop} \rangle\rangle \\
 &\quad \mid (- \vee -) \langle\langle \text{Prop} \times \text{Prop} \rangle\rangle \\
 &\quad \mid (- \Rightarrow -) \langle\langle \text{Prop} \times \text{Prop} \rangle\rangle \\
 &\quad \mid (- \Leftrightarrow -) \langle\langle \text{Prop} \times \text{Prop} \rangle\rangle
 \end{aligned}$$

Note that the definition of *Prop* depends on the definition of *PropVar*, the set of propositional variables. There are two *constant* propositions, namely *true* and *false*. Any propositional variable defines a proposition. Propositions are built up recursively from the constants and variables using *logical connectives*. Here I use logical connective symbols defined by Z rather than those defined by Lemmon.

### 2.2 PropVar

The separation of the form of a statement from any given subject matter is accomplished by the use of *propositional variables* that stand for arbitrary statements. Let *PropVar* denote the free type of all propositional variables.

**Remark.** *There are a countable infinity of propositional variables, e.g. P, P', P'', ...*

$$\text{PropVar} \twoheadrightarrow \mathbb{N} \neq \emptyset$$

### 2.3 $P \backslash \text{prop} P, Q \backslash \text{prop} Q, R \backslash \text{prop} R, S \backslash \text{prop} S, \text{ and } T \backslash \text{prop} T$

Traditionally, arbitrary statements are represented by single letters such as  $P, Q, R, S$ , and  $T$ .

**Remark.** *Each of  $P, Q, R, S$ , and  $T$  is a propositional variable.*

$$\{P, Q, R, S, T\} \subset \text{PropVar}$$

### 2.4 $' \backslash \text{propPrime}$

Typical propositions contain a small number of distinct statements, in which case the letters can be used. If more statements occur then the letters are decorated with one or more primes, e.g.  $P', Q''$ , etc.

**Example.**  $P'$  and  $Q''$  are propositional variables.

$$P' \in \text{PropVar}$$

$$Q'' \in \text{PropVar}$$

**Remark.** *Appending a prime to a propositional variable is an injection from  $\text{PropVar}$  to  $\text{PropVar}$ .*

$$(-') \in \text{PropVar} \rightarrow \text{PropVar}$$

### 2.5 $\text{true} \backslash \text{trueProp}$ and $\text{false} \backslash \text{falseProp}$

Let  $\text{true}$  denote the proposition that is true in all contexts and let  $\text{false}$  denote the proposition that is false in all contexts. The propositional  $\text{true}$  and  $\text{false}$  are said to be *constant* because they do not depend on the context.

### 2.6 $\text{var} \backslash \text{varProp}$

Let  $V$  be a propositional variable. Let  $\text{var}(V)$  denote the proposition defined by  $V$ .

**Remark.** *The sets  $\text{PropVar}$  and  $\text{Prop}$  are different types. The expression  $P$  is a propositional variable and the expression  $\text{var}(P)$  is a proposition.*

$$P \in \text{PropVar}$$

$$\text{var}(P) \in \text{Prop}$$

### 2.7 $\neg \backslash \text{notProp}$

Let  $A$  be a proposition. Let  $\neg A$  denote the *negation* of  $A$ .

**Example.**  $\neg(\text{var } P)$  is a proposition.

$$\neg(\text{var } P) \in \text{Prop}$$

## 2.8 $\neg$ \notPropV

We can simplify the notation for negating a proposition defined by a propositional variable by defining a function that directly negates the propositional variable and produces a proposition.

$$\left| \begin{array}{l} \neg : PropVar \rightarrow Prop \\ \hline \forall V : PropVar \bullet \\ \quad \neg V = \neg (\text{var } V) \end{array} \right.$$

## 2.9 $\wedge$ \andProp

Let  $A$  and  $B$  be propositions. Let  $A \wedge B$  denote the *conjunction* of  $A$  and  $B$ .

## 2.10 $\wedge$ \andPropVP, $\wedge$ \andPropPV, and $\wedge$ \andPropVV

We can simplify the notation for conjunctions involving propositions defined by propositional variables as follows.

$$\left| \begin{array}{l} \_ \wedge \_ : PropVar \times Prop \rightarrow Prop \\ \_ \wedge \_ : Prop \times PropVar \rightarrow Prop \\ \_ \wedge \_ : PropVar \times PropVar \rightarrow Prop \\ \hline \forall V : PropVar; A : Prop \bullet \\ \quad V \wedge A = (\text{var } V) \wedge A \\ \\ \forall A : Prop; V : PropVar \bullet \\ \quad A \wedge V = A \wedge (\text{var } V) \\ \\ \forall V, W : PropVar \bullet \\ \quad V \wedge W = (\text{var } V) \wedge (\text{var } W) \end{array} \right.$$

## 2.11 $\vee$ \orProp

Let  $A$  and  $B$  be propositions. Let  $A \vee B$  denote the *disjunction* of  $A$  and  $B$ .

## 2.12 $\vee$ \orPropVP, $\vee$ \orPropPV, and $\vee$ \orPropVV

We can simplify the notation for disjunctions involving propositions defined by propositional variables as follows.

$$\begin{array}{|l}
\hline
\_ \vee \_ : PropVar \times Prop \multimap Prop \\
\_ \vee \_ : Prop \times PropVar \multimap Prop \\
\_ \vee \_ : PropVar \times PropVar \multimap Prop \\
\hline
\forall V : PropVar; A : Prop \bullet \\
\quad V \vee A = (\text{var } V) \vee A \\
\\
\forall A : Prop; V : PropVar \bullet \\
\quad A \vee V = A \vee (\text{var } V) \\
\\
\forall V, W : PropVar \bullet \\
\quad V \vee W = (\text{var } V) \vee (\text{var } W)
\end{array}$$

### 2.13 $\Rightarrow$ \impliesProp

Let  $A$  and  $B$  be propositions. Let  $A \Rightarrow B$  denote the *implication* of  $A$  and  $B$ .

### 2.14 $\Rightarrow$ \impliesPropVP, $\Rightarrow$ \impliesPropPV, and $\Rightarrow$ \impliesPropVV

We can simplify the notation for implications involving propositions defined by propositional variables as follows.

$$\begin{array}{|l}
\hline
\_ \Rightarrow \_ : PropVar \times Prop \multimap Prop \\
\_ \Rightarrow \_ : Prop \times PropVar \multimap Prop \\
\_ \Rightarrow \_ : PropVar \times PropVar \multimap Prop \\
\hline
\forall V : PropVar; A : Prop \bullet \\
\quad V \Rightarrow A = (\text{var } V) \Rightarrow A \\
\\
\forall A : Prop; V : PropVar \bullet \\
\quad A \Rightarrow V = A \Rightarrow (\text{var } V) \\
\\
\forall V, W : PropVar \bullet \\
\quad V \Rightarrow W = (\text{var } V) \Rightarrow (\text{var } W)
\end{array}$$

### 2.15 $\Leftrightarrow$ \equivProp

Let  $A$  and  $B$  be propositions. Let  $A \Leftrightarrow B$  denote the *equivalence* of  $A$  and  $B$ .

### 2.16 $\Leftrightarrow$ \equivPropVP, $\Leftrightarrow$ \equivPropPV, and $\Leftrightarrow$ \equivPropVV

We can simplify the notation for equivalences involving propositions defined by propositional variables as follows.

$- \Leftrightarrow - : PropVar \times Prop \multimap Prop$
$- \Leftrightarrow - : Prop \times PropVar \multimap Prop$
$- \Leftrightarrow - : PropVar \times PropVar \multimap Prop$
$\forall V : PropVar; A : Prop \bullet$
$V \Leftrightarrow A = (\text{var } V) \Leftrightarrow A$
$\forall A : Prop; V : PropVar \bullet$
$A \Leftrightarrow V = A \Leftrightarrow (\text{var } V)$
$\forall V, W : PropVar \bullet$
$V \Leftrightarrow W = (\text{var } V) \Leftrightarrow (\text{var } W)$

### 3 Proofs

Lemmon has a nice way of presenting proofs. Here's an example.

1	$P \Rightarrow Q, P \vdash Q$
1	(1) $P \Rightarrow Q$ A
2	(2) $P$ A
1,2	(3) $Q$ 1,2 MPP

A proof consists of a *sequent* to be proved and a finite sequence of *lines*. The lines are labelled by consecutive natural numbers, starting at 1. Each line contains a proposition and the application of the *proof rule* used to add it to the proof. A proof rule is either an *assumption* or a *derivation*. The *rule of assumption* allows the addition of any proposition. The *rules of derivation* allow the addition of a *conclusion* derived from *premises* that have been previously added. A premise is therefore either an assumption or the conclusion of a previous deduction. Every line of the proof can therefore be traced back to a finite, possibly empty, set of assumptions upon which it ultimately *depends*.

#### 3.1 ProofRule

Let *ProofRule* denote the set of proof rule applications.

$$ProofRule ::= A \mid MPP \langle \langle \mathbb{N}_1 \times \mathbb{N}_1 \rangle \rangle$$

**Example.**  $A$  and  $MPP(1, 2)$  are proof rules applications.

$$A \in ProofRule$$

$$MPP(1, 2) \in ProofRule$$

### 3.2 *ProofLine*

A proof line consists of a finite, possibly empty, set of assumptions, a proposition, and the proof rule application used to derive the proposition. Let *ProofLine* denote the set of all proof lines.

<i>ProofLine</i>	_____
<i>assumptions</i> :	$\mathbb{F} \mathbb{N}_1$
<i>prop</i> :	<i>Prop</i>
<i>rule</i> :	<i>ProofRule</i>

### 3.3 *ProofLineTuple*

Let *ProofLineTuple* denote the tuple formed by the components of a proof line.

$$ProofLineTuple == \mathbb{F} \mathbb{N}_1 \times Prop \times ProofRule$$

**Example.** Line 3 of proof 1 corresponds to the following proof line.

$$(\{1, 2\}, \text{var}(Q), \text{MPP}(1, 2)) \in ProofLineTuple$$

### 3.4 tuple \proofLineTuple

Let tuple map a proof line to its tuple.

$$\text{tuple} == (\lambda ProofLine \bullet (assumptions, prop, rule))$$

**Remark.** The mapping from proof lines to tuples is a bijections.

$$\text{tuple} \in ProofLine \rightsquigarrow ProofLineTuple$$

### 3.5 *Argument*

An *argument* is a finite sequence of proof lines. Let *Argument* denote the set of all arguments.

$$Argument == \text{seq } ProofLineTuple$$

### 3.6 *Proof<sub>1</sub>*

Let *Proof<sub>1</sub>* denote the argument defined by proof 1.

$$Proof_1 == \begin{aligned} &\{1 \mapsto (\{1\}, P \Rightarrow Q, A), \\ &2 \mapsto (\{2\}, \text{var}(P), A), \\ &3 \mapsto (\{1, 2\}, \text{var}(Q), \text{MPP}(1, 2))\} \end{aligned}$$

**Example.**  $Proof_1$  is an argument.

$$Proof_1 \in Argument$$

### 3.7 *ArgumentLine*

A argument is said to be *sound* or *valid* if each proof line in the argument obeys the rules of derivation. When assessing the validity of a proof line, we need to specify its line number within the argument and examine its contents. Let *ArgumentLine* denote the set of all arguments and line numbers within it.

<i>ArgumentLine</i>	_____
<i>argument</i> : <i>Argument</i>	
<i>lineNumber</i> : $\mathbb{N}_1$	
<i>lineNumber</i> $\in \text{dom } argument$	

- The line number corresponds to a proof line within the argument.

**Example.** *Line 3 in contained in proof 1.*

**let** *argument* ==  $Proof_1$ ; *lineNumber* == 3 •  
*ArgumentLine*

### 3.8 *RuleOfAssumption*

A proof line that uses the rule of assumption A in an argument is sound if it depends only on itself. Let *RuleOfAssumption* denote the set of all argument lines that use the rule of assumption soundly.

<i>RuleOfAssumption</i>	_____
<i>ArgumentLine</i>	
<i>P</i> : <i>Prop</i>	
$argument(lineNumber) = (\{lineNumber\}, P, A)$	

- The line depends only on itself and the rule is the rule of assumptions A.

### 3.9 *AssumptionArgumentLine*

Let *AssumptionArgumentLine* denote the set of all argument lines that are sound applications of the rule of assumption.

$$AssumptionArgumentLine \hat{=} RuleOfAssumption \upharpoonright ArgumentLine$$



**Example.** Lines 1 and 2 of *Proof<sub>1</sub>* are sound applications of the rule of assumption.

```

let argument == Proof1 •
    ∀ lineNumber : {1, 2} •
        AssumptionArgumentLine

```

### 3.10 RuleOfMPP

A proof line that uses the rule  $\text{MPP}(i, j)$  in an argument is sound if  $i$  and  $j$  are lines that precede it, the proposition on line  $i$  is an implication, the proposition on line  $j$  is the antecedent of the implication, the proposition of the proof line is the consequent of the implication, and the proof line's assumptions is the union of the assumptions of lines  $i$  and  $j$ . Let *RuleOfMPP* denote the set of all argument lines that use the rule of MPP soundly.

<i>RuleOfMPP</i>	
<i>ArgumentLine</i>	
$i, j : \mathbb{N}_1$	
<i>ProofLine<sub>1</sub></i>	
<i>ProofLine<sub>2</sub></i>	
$P, Q : \text{Prop}$	
$i < \text{lineNumber} \wedge j < \text{lineNumber}$	
$\text{argument}(i) = (\text{assumptions}_1, P \Rightarrow Q, \text{rule}_1)$	
$\text{argument}(j) = (\text{assumptions}_2, P, \text{rule}_2)$	
$\text{argument}(\text{lineNumber}) = (\text{assumptions}_1 \cup \text{assumptions}_2, Q, \text{MPP}(i, j))$	

### 3.11 MPPArgumentLine

Let *MPPArgumentLine* denote the set of all argument lines that are sound applications of the rule of MPP.

$$\text{MPPArgumentLine} \hat{=} \text{RuleOfMPP} \upharpoonright \text{ArgumentLine}$$

**Example.** Line 3 of *Proof<sub>1</sub>* is a sound application of the rule of MPP.

```

let argument == Proof1;
    lineNumber == 3 •
        MPPArgumentLine

```

### 3.12 *SoundArgumentLine*

An argument line is sound if it is a sound application of one of the rules of derivation. Let *SoundArgumentLine* denote the set of all sound argument lines.

$$\begin{aligned} \text{SoundArgumentLine} &\hat{=} \\ &\text{AssumptionArgumentLine} \vee \\ &\text{MPPArgumentLine} \end{aligned}$$

### 3.13 *Proof*

A proof is an argument in which every line is sound. Let *Proof* denote the set of proofs.

$$\begin{aligned} \text{Proof} = & \{ \text{argument} : \text{Argument} \mid \\ & \forall \text{lineNumber} : \text{dom argument} \bullet \\ & \text{SoundArgumentLine} \} \end{aligned}$$

**Example.** *Proof<sub>1</sub>* is a proof.

$$\text{Proof}_1 \in \text{Proof}$$

## 4 The Curry-Howard Correspondence

The *Curry-Howard Correspondence* is an interpretation of propositions and proofs in terms of types. To each proposition there corresponds a type. The inhabitants of the type that corresponds to a proposition are proofs of that proposition. The logical connectives that build up propositions correspond to type constructors.

To illustrate the correspondence, consider the implication logical connective  $P \Rightarrow Q$ . It corresponds to the type constructor  $P \rightarrow Q$ . A proof  $f$  of  $P \Rightarrow Q$  corresponds to a function that maps any proof of  $P$  to some proof of  $Q$ .

Of course,  $P \rightarrow Q$  is not a type in  $\mathbf{Z}$ . It is a subset of the type  $\mathbb{P}(P \times Q)$ . Nevertheless, we'll press on and pretend that it is a  $\mathbf{Z}$  type for now.

The rules of derivation correspond to the construction of a proof from assumptions. For example the rule of MPP corresponds to function application. Consider *Proof<sub>1</sub>*. It corresponds to the following.

$$\begin{array}{|l} \text{CHProof1}[P, Q] \text{-----} \\ f : P \rightarrow Q \\ x : P \\ y : Q \\ \hline y = f(x) \end{array}$$